

Chương 6

HOẠT ĐỘNG NGẮT INTERRUPT

cuu duong than cong . com

cuu duong than cong . com

I. KHÁI NIỆM

- **Ngắt** là sự xảy ra của một điều kiện(sự kiện) làm cho chương trình bị tạm ngưng và điều kiện được phục vụ bởi một chương trình khác **ISR (Interrupt service routine** - gọi là chương trình con phục vụ ngắt).
- Khi một ngắt xuất hiện, việc thực thi chương trình chính tạm thời bị ngừng và CPU thực hiện việc rẽ nhánh đến trình con phục vụ ngắt.
- Sau khi việc thực thi ISR kết thúc, chương trình chính lại tiếp tục tại nơi bị tạm dừng.

cuu duong than cong . com

SỰ KHÁC NHAU GIỮA ISR VÀ CHƯƠNG TRÌNH CON

- **Giống:** CPU cùng rẽ nhánh đến chương trình khác và trở về chương trình chính.
- **Khác:** CPU không biết khi nào thì sẽ thực thi ISR. Lệnh kết thúc của **ISR** là **RETI**, còn **chương trình con** là **RET**.

cuu duong than cong . com

II. TỔ CHỨC NGẮT CỦA 8051

- 2 ngắt ngoài
- 2 ngắt do bộ định thời
- 1 ngắt do port nối tiếp
- *Ngắt Timer2 (đối với 8052)*

Nguyên nhân xảy ra ngắt

- Khi có 1 cạnh xuống xảy ra trên chân /INT0 thì **ngắt ngoài 0** xuất hiện.
- Khi có 1 cạnh xuống xảy ra trên chân /INT1 thì **ngắt ngoài 1** xuất hiện.
- Khi **Timer0** tràn thì **ngắt timer0** xuất hiện.
- Khi **Timer1** tràn thì **ngắt timer1** xuất hiện.
- Khi việc **phát** được thực thi (bộ đếm phát rỗng) hoặc việc **thu** đã xong (bộ đếm thu đầy) thì sẽ xảy ra **ngắt do port nối tiếp**.

Cho phép và không cho phép ngắt

Mỗi một nguyên nhân ngắt được cho phép hoặc không cho phép riêng rẽ thông qua thanh ghi chức năng đặc biệt định địa chỉ bit là thanh ghi IE có địa chỉ byte là 0A8H.

EA	-	ET2	ES	ET1	EX1	ET0	EX0
----	---	-----	----	-----	-----	-----	-----

Thanh ghi cho phép ngắt IE

- **EA** : Cho phép/không cho phép toàn cục
- **-** : Không sử dụng
- **ET2**: Cho phép Ngắt do bộ định thời 2 (8052)
- **ES**: Cho phép Ngắt do port nối tiếp
- **ET1**: Cho phép Ngắt do bộ định thời 1
- **EX1**: Cho phép Ngắt ngoài 1
- **ET0**: Cho phép Ngắt do bộ định thời 0
- **EX0**: Cho phép Ngắt ngoài 0

1 = Cho phép ; 0 = Không Cho phép

Ưu tiên ngắt

Mỗi một nguyên nhân ngắt được lập trình riêng rẽ để có một trong hai mức ưu tiên thông qua thanh ghi chức năng đặc biệt, thanh ghi ưu tiên ngắt IP có địa chỉ byte là 0B8H.

-	-	PT2	PS	PT1	PX1	PT0	PX0
---	---	-----	----	-----	-----	-----	-----

Thanh ghi IP (Interrupt Priority)

- **PT2** : Ưu tiên cho ngắt bộ định thời 2.
- **PS**: Ưu tiên cho ngắt port nối tiếp
- **PT1**: Ưu tiên cho ngắt bộ định thời 1
- **PX1**: Ưu tiên cho ngắt ngoài 1
- **PT0**: Ưu tiên cho ngắt bộ định thời 0
- **PX0**: Ưu tiên cho ngắt ngoài 0

1 = mức cao hơn; 0 = mức thấp hơn

Chuỗi vòng

- Nếu có 2 ngắt xảy ra đồng thời thì ngắt nào có mức ưu tiên cao hơn sẽ được thực thi trước.
- Nếu một ngắt có mức ưu tiên cao xuất hiện trong khi một ngắt có mức thấp đang thực thi thì trình phục vụ ngắt cho ngắt có mức ưu tiên thấp phải tạm dừng. Ta không thể tạm dừng một chương trình phục vụ ngắt có mức ưu tiên cao.
- Nếu 2 ngắt có cùng mức ưu tiên xảy ra đồng thời, **chuỗi vòng** cố định sẽ xác định ngắt nào được phục vụ trước.
- Chuỗi vòng này sẽ là: ngắt ngoài 0, ngắt do bộ định thời 0, ngắt ngoài 1, ngắt do bộ định thời 1, ngắt do port nối tiếp, ngắt do bộ định thời 2.

III. XỬ LÝ NGẮT

Khi có một ngắt xuất hiện và được CPU chấp nhận, chương trình chính bị ngắt và các thao tác sau đây xảy ra:

- Cờ ngắt tương ứng được tự động đặt lên 1.
- CPU hoàn tất việc thực thi lệnh hiện hành.
- Nội dung của PC được cất vào stack.
- Bộ đếm chương trình PC được nạp địa chỉ vector của trình phục vụ ngắt ISR.
- Cờ ngắt tự động được xóa, ngoại trừ RI & TI (và TF2 & EXF2 của 8052).
- ISR được thực thi để đáp ứng công việc của ngắt. Việc thực thi ISR kết thúc khi gặp lệnh RETI (trở về từ một trình phục vụ ngắt).
- Bộ đếm chương trình PC lấy lại giá trị cũ từ stack và chương trình chính được tiếp tục ở nơi bị tạm ngưng.

Các cờ ngắt

Khi có một sự kiện xảy ra:

→ **Cờ ngắt tương ứng sẽ được set lên 1** → **xảy ra ngắt.**

Đối với 8051 có 6 cờ ngắt tương ứng: **IE0, IE1, TF0, TF1, TI, và RI**

- Khi có 1 cạnh xuống xảy ra trên chân **/INT0** → **IE0 = 1** → **ngắt ngoài 0** xuất hiện.
- Khi có 1 cạnh xuống xảy ra trên chân **/INT1** → **IE1 = 1** → **ngắt ngoài 1** xuất hiện.
- Khi **Timer0** tràn → **TF0 = 1** → **ngắt timer0** xuất hiện.
- Khi **Timer1** tràn → **TF1 = 1** → **ngắt timer1** xuất hiện.
- Khi việc **phát** được thực thi (bộ đếm phát rỗng) hoặc việc **thu** đã xong (bộ đếm thu đầy) → **TI = 1** hoặc **RI=1** → sẽ xảy ra **ngắt do port nối tiếp.**

Các vector ngắt

Khi một ngắt được chấp nhận, giá trị được nạp cho bộ đếm chương trình PC được gọi là vector ngắt.

Vector ngắt là địa chỉ bắt đầu của trình phục vụ ngắt của nguyên nhân ngắt tương ứng. Các vector ngắt được cho ở bảng sau:

NGẮT	Cờ ngắt	Địa chỉ vector ngắt
Reset hệ thống	RST	0000H
Ngắt ngoài 0	IE0	0003H
Ngắt Timer 0	TF0	000BH
Ngắt ngoài 1	IE1	0013H
Ngắt Timer 1	TF1	001BH
Ngắt port nối tiếp	RI or TI	0023H
Ngắt Timer 2	TF 2 or EXF2	002BH

III. THIẾT KẾ CHƯƠNG TRÌNH SỬ DỤNG NGẮT

Khuôn mẫu đề nghị cho một chương trình được thực thi độc lập có sử dụng ngắt như sau:

```
ORG 0000H    ;điểm nhập sau khi Reset
LJMP MAIN
...          ;các điểm nhập của ISR
...
...          cuu duong than cong . com
ORG 0030H    ;điểm nhập của chương trình chính
MAIN: ...    ; chương trình chính bắt đầu
...
END          cuu duong than cong . com
```

Các trình phục vụ ngắt kích thước nhỏ

Khi các chương trình phục vụ ngắt có kích thước không dài quá 8byte và nếu chỉ có một nguyên nhân ngắt được dùng, giả sử là ngắt do bộ định thời 0, thì khuôn mẫu sau có thể được sử dụng

```
ORG 0000H      ;Reset
LJMP MAIN
ORG 000BH      ;Điểm nhập ngắt do Timer 0
T0ISR:...      ;bắt đầu ISR cho timer0
...
RETI           ;trở về từ chương trình chính
MAIN: ...      ;chương trình chính
...
END
```

Các trình phục vụ ngắt kích thước lớn

Khi các chương trình phục vụ ngắt có kích thước lớn và có nhiều nguồn ngắt được yêu cầu thì khuôn mẫu sau có thể được sử dụng

```
ORG 0000H           ;Reset
LJMP MAIN
ORG 0003H           ;Interrupt vector of External 0
LJMP E0ISR
ORG 000BH           ;Interrupt vector of Timer 0
LJMP T0ISR
ORG 0013H           ;Interrupt vector of External 1
LJMP E1ISR
ORG 001BH           ;Interrupt vector of Timer 1
LJMP T1ISR
ORG 0023H           ;Interrupt vector of serial port
LJMP SPISR
ORG 0030H
MAIN: ...
      SJMP $
E0ISR: ...
      RETI
T0ISR: ...
      RETI
      ...
      END
```

NGẮT TIMER

Timer 8bit

Ví dụ

Viết chương trình sử dụng Timer0 và các ngắt để tạo ra một sóng vuông tần số 10KHz trên chân P1.0 (XTAL 12MHz).

```
ORG 0000H      ;Reset
LJMP MAIN
ORG 000BH      ;Vector ngắt Timer 0
T0ISR: CPL P1.0
      RETI
      ORG 0030H
MAIN: MOV TMOD,#02H
      MOV TH0,#-50
      SETB TR0
      MOV IE,#82H
      SJMP $
      END
```

Lưu ý: Cờ tràn TF0 không cần được xóa bởi phần mềm do khi các ngắt được cho phép, cờ này tự động được xóa bởi phần cứng khi CPU trở tới trình phục vụ ngắt.

Timer 16bit

Ví dụ

Viết chương trình sử dụng Timer 0 và các ngắt để tạo ra một sóng vuông tần số 1KHz trên chân P1.1 (XTAL 12MHz).

```
ORG 0000H
LJMP CTCHINH
ORG 000BH
LJMP CTNGATTIMER0
ORG 0030H
CTCHINH:
MOV TMOD,#01H
SETB EA
SETB ET0
SETB TF0 ; Ép ngắt
SJMP $

CTNGATTIMER0:
CPL P1.1
CLR TR0
MOV TH0,#HIGH(-500)
MOV TL0,#LOW(-500)
SETB TR0
RETI
END
```

Lưu ý: Khác với mode 2, các thanh ghi TL0/TH0 phải được nạp lại sau mỗi lần tràn timer, TF0 được set bằng 1 trong chương trình chính bằng phần mềm để buộc phải có 1 ngắt ban đầu ngay khi các ngắt được cho phép.

NGẮT PORT NỘI TIẾP

- Các ngắt do port nối tiếp xuất hiện khi cờ ngắt phát TI hoặc cờ ngắt thu RI được set bằng 1.
- Cờ ngắt do port nối tiếp gây ra không được xóa bởi phần cứng khi CPU trở tới trình phục vụ ngắt.
- Nguyên nhân ngắt phải được xác định trong trình phục vụ ngắt và cờ tạo ra ngắt được xóa bởi phần mềm.

Ví dụ

Viết chương trình sử dụng ngắt để liên tục phát đi tập mã ASCII (ngoại trừ các mã điều khiển) đến 1 thiết bị đầu cuối nối với 8051 qua port nối tiếp (1200 baud, XTAL 12MHz).

Bảng mã ASCII 7 bit bao gồm 95 mã đồ họa (từ 20H đến 7EH) và 33 mã điều khiển (từ 00H đến 1FH và 7EH)

ORG 0000H	
LJMP MAIN	
ORG 0023H	;điểm vào ngắt cổng nối tiếp
LJMP SPISR	
ORG 0030H	
MAIN: MOV TMOD,#20H	;Timer1, mode 2
MOV TH1,#-26	;giá trị nạp lại 1200 baud
SETB TR1	
MOV SCON,#42H	;SIO mode1, set TI để ép ngắt ban đầu
MOV A,#20H	;gửi ký tự trắng
MOV IE,#90H	
SJMP \$	;lặp tại chỗ
SPISR: CJNE A,#7FH,SKIP	;nếu kết thúc bộ mã ASCII
MOV A,#20H	;reset về ký tự trắng
SKIP: MOV SBUF,A	;gửi ký tự đến port nối tiếp
INC A	;ký tự kế
CLR TI	;xóa cờ ngắt phát
RETI	
END	

NGẮT NGOÀI

- Các ngắt ngoài xảy ra khi có mức thấp hoặc có cạnh âm trên chân $\overline{INTx}$ ($x=0,1$) của 8051.
- Các cờ tạo ra các ngắt này là IE0 và IE1 của thanh ghi TCON.
- Các cờ này được xóa bởi phần cứng khi CPU trở đến trình phục vụ ngắt nếu chỉ nếu ngắt thuộc loại tác động cạnh. Nếu ngắt thuộc loại tác động mức, nguyên nhân ngắt ngoài sẽ điều khiển mức của cờ thay vì là phần cứng trên chip.
- Việc chọn các ngắt loại tác động cạnh hay mức được lập trình thông qua các bit IT0 và IT1 của thanh ghi TCON.


Ví dụ

Sử dụng các ngắt để thiết kế bộ điều khiển lò nung sao cho nhiệt độ được duy trì ở $20^{\circ}\text{C} \pm 1^{\circ}\text{C}$.


Bộ cảm biến nhiệt độ được nối với $\overline{INT0}$ và $\overline{INT1}$, cung cấp các tín hiệu $\overline{HOT}$ và $\overline{COLD}$ như sau:

$$\overline{HOT} = 0 \text{ nếu } T > 21^{\circ}\text{C} \text{ và } \overline{COLD} = 0 \text{ nếu } T < 19^{\circ}\text{C}$$

Cấu hình phần cứng và giản đồ thời gian cho ví dụ điều khiển lò



(a)



(b)

```

ORG 0000H
LJMP MAIN
ORG 0003H
E0ISR: CLR P1.7           ;tắt lò
RETI
ORG 0013H
E1ISR: SETB P1.7         ;mở lò
RETI
ORG 0030H
MAIN: MOV IE,#85H        ;Cho phép các ngắt ngoài
SETB IT0                 ;kích khởi cạnh âm
SETB IT1                 ;
SETB P1.7                ;mở lò
JB P3.2,SKIP             ;nếu T>21°C,
CLR P1.7                 ; tắt lò
SKIP: SJMP $              ;không làm gì
END

```